Funktionales Programmieren
(Praktische Informatik 3)

Berthold Hoffmann
Studiengang Informatik
Universität Bremen

Winter 03/04

Vorlesung vom 03.11.2002: Listenumschreibungen, Polymorphie und Rekursion

Inhalt

- Letzten Montag
  - Basisdatentypen, Tupel und Listen
  - Funktionsdefinition mit Gleichungen und pattern matching
- Heute: Funktionsdefinition durch
  - Listenumschreibung
  - primitive Rekursion bzw. nicht-primitive Rekursion
- Neue Sprachkonzepte:
  - Polymorphie — eine Erweiterung des Typkonzeptes
  - Lokale Definitionen
- Vordefinierte Funktionen auf Listen

Listenumschreibungen

- Ein Schema für Funktionen auf Listen:
  - Eingabe generiert Elemente,
  - die auf Erfüllung eines Prädikats getestet und
  - zu einem Ergebns transformiert werden

- Ein Schema für Funktionen auf Listen:
  - Eingabe generiert Elemente,
  - die auf Erfüllung eines Prädikats getestet und
  - zu einem Ergebns transformiert werden
- Beispiel Palindrom:
  - alle Buchstaben im String `str` zu Kleinbuchstaben.

    ```
    [ toLower c | c <- str ]
    ```

  - Alle Buchstaben aus `str` herausfiltern:

    ```
    [ c | c <- str, isAlpha c ]
    ```

  - Beides zusammen:

    ```
    [ toLower c | c<- str, isAlpha c]
    ```

- Allgemeine Form:

```
[E |c1<- L1, ..., ck<- Lk, P1, ..., Pn ]
```

- Generator mit pattern matching:

```
addPairwise :: [(Int, Int)]-> [Int]
addPairwise ls = [ x+y | (x, y) <- ls ]
```

- Ähnlich zur Mengenschreibweise (nicht gleich!)

$$\{E \mid c_1 \in M_1, ..., c_k \in M_k, P_1, ..., P_n\}$$

  - Mengen sind ungeordnet und Doubletten-frei
  - Listen sind geordnet und können Doubletten enthalten

- Beispiel Quicksort:
  - Zerlege Liste in Elemente kleiner gleich und größer dem ersten,
  - sortiere Teilstücke,
  - konkateniere Ergebnisse.

- Beispiel Quicksort:
  - Zerlege Liste in Elemente kleiner gleich und größer dem ersten,
  - sortiere Teilstücke,
  - konkateniere Ergebnisse.

```
qsort :: [Int]-> [Int]
qsort [] = []
qsort (x:xs) = qsort [ y | y <- xs, y<= x ]
               ++ [x] ++
               qsort [ y | y <- xs, y > x ]
```

Beispiel: Eine Bücherei

- Problem: Modellierung einer Bücherei
- Datentypen:
  - Ausleihende Personen
  - Bücher
  - Zustand der Bücherei: ausgeliehene Bücher, Ausleiher

```
type Person   = String
type Book     = String
type DBase = [(Person, Book)]
```

- Buch ausleihen und zurückgeben:

```
makeLoan :: DBase -> Person -> Book -> DBase
makeLoan dBase pers bk = [(pers,bk)] ++ dBase
```

  - Benutzt `(++)` zur Verkettung von `DBase`

```
returnLoan :: DBase -> Person -> Book -> DBase
returnLoan dBase pers bk
    = [ loan | loan <- dBase, loan /= (pers,bk) ]
```

- Suchfunktionen: Wer hat welche Bücher ausgeliehen (Test)

```
books :: DBase -> Person -> [Book]
books db who = [ book | (pers,book)<- db,
                        pers== who ]
```

Polymorphie — jetzt oder nie.

- Definition von `(++)`:

```
(++) :: [DBase]-> [DBase]-> [DBase]
[] ++ ys      = ys
(x:xs) ++ ys = x:(xs++ ys)
```

- Verketten von Strings:

Polymorphie — jetzt oder nie.

- Definition von `(++)`:

```
(++) :: [DBase]-> [DBase]-> [DBase]
[] ++ ys      = ys
(x:xs) ++ ys = x:(xs++ ys)
```

- Verketten von Strings:

```
(++) :: String-> String-> String
[] ++ ys      = ys
(x:xs) ++ ys = x:(xs++ ys)
```

- Gleiche Definition, aber unterschiedlicher Typ!

  $\Longrightarrow$ Zwei Instanzen einer allgemeineren Definition.

- Polymorphie erlaubt Abstraktion von Typen:

```
(++) :: [a]-> [a]-> [a] -- for every type a
[] ++ ys      = ys
(x:xs) ++ ys = x:(xs++ ys)
```

  a ist hier eine Typvariable. (Auf Papier: $\alpha, \beta, \ldots$)

- Definition wird bei Anwendung instantiiert:

```
[3,5,57] ++ [39, 18]      "hi" ++ "ho"
```

  aber nicht

```
[True, False] ++ [18, 45]
```

- Weitere Beispiele:
  - Länge einer Liste:

```
length :: [a] -> Int
length []     = 0
length (x:xs) = 1+ length xs
```

  - Verschachtelte Listen "flachklopfen":

```
concat :: [[a]]-> [a]
concat []     = []
concat (x:xs) = x ++ (concat xs)
```

  - Kopf und Rest einer nicht-leeren Liste:

```
head :: [a]-> a
head (x:xs) = x
```

```
tail :: [a]-> [a]
tail (x:xs) = xs
```

Undefiniert für leere Liste.

Übersicht: vordefinierte Funktionen auf Listen

`:`	`a-> [a]-> [a]`	Element vorne anfügen
`++`	`[a]-> [a]-> [a]`	Verketten
`!!`	`[a]-> Int-> a`	`n`-tes Element selektieren
`concat`	`[[a]]-> [a]`	"flachklopfen"
`length`	`[a]-> Int`	Länge
`head`, `last`	`[a]-> a`	Erstes/letztes Element
`tail`, `init`	`[a]-> [a]`	Rest (hinterer/vorderer)
`replicate`	`Int-> a-> [a]`	Erzeuge `n` Kopien
`take`	`Int-> [a]-> [a]`	Nimmt erste `n` Elemente
`drop`	`Int-> [a]-> [a]`	Entfernt erste `n` Elemente
`splitAt`	`Int-> [a]-> ([a], [a])`	Spaltet an `n`-ter Position

`reverse`	`[a]-> [a]`	Dreht Liste um
`zip`	`[a]-> [b]-> [(a, b)]`	Macht aus Paar von Listen Liste von Paaren
`unzip`	`[(a, b)]-> ([a], [b])`	Macht aus Liste von Paaren Paar von Listen
`and`, `or`	`[Bool]-> Bool`	Konjunktion/Disjunktion
`sum`	`[Int]-> Int` (überladen)	Summe
`product`	`[Int]-> Int` (überladen)	Produkt

Siehe Thompson S. 91/92.

Palindrom zum letzten:

```
palindrom xs = (reverse l) == l where
             l = [toLower c| c<- xs, isAlpha c]
```

Lokale Definitionen

- Lokale Definitionen mit `where` und `let`:

```
f x y                          f x y
  | g1 = P                       | g1 = P
  | g2 = Q where                 | g2 = let v1 = M
           v1 = M                           v2 x = N x
           v2 x = N x                  in  Q
```

- `v1`, `v2`, . . . werden gleichzeitig definiert (Rekursion!);
- Namen `v1` und Parameter (`x`) überlagern andere;
- Es gilt die Abseitsregel (deshalb auf gleiche Einrückung der lokalen Definitionen achten);

Muster*(pattern)*

Muster*(pattern)*

- Funktionsparameter sind Muster (z.B. `head (x:xs) = x`)
  - Wert (`0` oder `True`)
  - Variable (`x`) - dazu paßt alles
    - ▷ Jede Variable darf links nur einmal auftreten.
  - namenloses Muster (`_`) - dazu paßt alles.
    - ▷ `_` darf links mehrfach, rechts gar nicht auftreten.
  - Tupel `(p1, p2, ... pn)` (`pi` sind wieder Muster)
  - leere Liste `[]`
  - nicht-leere Liste `ph:pl` (`ph, pl` sind wieder Muster)
  - `[p1,p2,...pn]` ist syntaktischer Zucker für `p1:p2:...pn:[]`

Primitive und allgemeine Rekursion

Primitive Rekursion auf Listen

- Primitive Rekursion vs. allgemeine Rekursion
- *Primitive Rekursion* : gegeben durch
  - eine Gleichung für die leere Liste
  - eine Gleichung für die nicht-leere Liste
- Beispiel:

```
sum :: [Int]-> Int
sum []     = 0
sum (x:xs) = x+ sum xs
```

- Weitere Beispiele: `length`, `concat`, `(++)`, . . .

- Auswertung:

```
sum [4,7,3]          ⇝  4 + 7 + 3 + 0
concat [A, B, C]     ⇝  A ++ B ++ C++ []
```

- Allgemeines Muster:

$$\texttt{f}[\texttt{x1}, \ldots, \texttt{xn}] = \texttt{x1} \otimes \texttt{x2} \otimes \ldots \otimes \texttt{xn} \otimes \texttt{neutr}$$

  - Neutraler Wert (für die leere Liste) `neutr :: b`
  - Rekursionsfunktion $\otimes$ `:: a -> b-> b`

- Entspricht einfacher Iteration (Schleife).

Nicht-primitive Rekursion

- *Allgemeine Rekursion* :
  - Rekursion über mehrere Argumente
  - Rekursion über andere Datenstruktur
  - Andere Zerlegung als Kopf und Rest
- Rekursion über mehrere Argumente:

```
zip :: [a]-> [b]-> [(a, b)]
zip [] _ = []
zip _ [] = []
zip (x:xs) (y:ys) = (x,y):(zip xs ys)
```

- Rekursion über ganzen Zahlen:

```
take :: Int-> [a]-> [a]
take 0 _  = []
take _ [] = []
take n (x:xs)
  | n > 0     = x: take (n-1) xs
  | otherwise = error "take: negative Argument"
```

- Quicksort:
  - zerlege Liste in Elemente kleiner gleich und größer dem ersten,
  - sortiere beide Teilstücke rekujrsiv, konkateniere Ergebnisse

- Mergesort:
  - teile Liste in der Hälfte,
  - sortiere Teilstücke, füge ordnungserhaltend zusammen.

```
msort :: [Int]-> [Int]
msort xs
   | length xs <= 1 = xs
   | otherwise = merge (msort front) (msort back) where
     (front, back) = splitAt ((length xs) `div` 2) xs
     merge :: [Int]-> [Int]-> [Int]
     merge [] x = x
     merge y [] = y
     merge (x:xs) (y:ys)
        | x<= y      = x:(merge xs (y:ys))
        | otherwise = y:(merge (x:xs) ys)
```

Beispiel: das n-Königinnen-Problem

- Problem: n Königinnen auf $n \times n$-Schachbrett
- Spezifikation:
  - Position der Königinnen

    ```
    type Pos = (Int, Int)
    ```

  - Eingabe: Anzahl Königinnen, Rückgabe: Liste von Lösungen

    ```
    queens :: Int-> [[Pos]]
    ```

- Rekursive Formulierung:
  - Keine Königin— kein Problem.
  - Lösung für n Königinnen: Lösung für $n-1$ Königinnen, und n-te Königin so stellen, dass keine andere sie bedroht.
  - n-te Königin muß in n-ter Spalte plaziert werden.

- Hauptfunktion:

```
queens num = qu num where
  qu :: Int-> [[Pos]]
  qu n | n == 0  = [[]]
       | otherwise =
            [ p++ [(n, m)] | p <- qu (n-1),
                             m <- [1.. num],
                             safe p (n, m)]
```

  - `[n..m]`: Liste der Zahlen von `n` bis `m`
  - Mehrere Generatoren in Listenumschreibung.
  - Rekursion über Anzahl der Königinnen.

- Sichere neue Position:
  - Neue Position ist sicher, wenn sie durch keine anderen bedroht wird:

```
safe :: [Pos]-> Pos-> Bool
safe others nu =
    and [ not (threatens other nu)
                        | other <- others ]
```

  - Verallgemeinerte Konjunktion `and :: [Bool]-> Bool`
- Gegenseitige Bedrohung: (Test)
  - Bedrohung wenn in gleicher Zeile, oder Diagonale. (Spalte nicht!)

```
threatens :: Pos-> Pos-> Bool
threatens (i, j) (m, n) =
    (j== n) || (i+j == m+n) || (i-j == m-n)
```

Zusammenfassung

- Schemata für Funktionen über Listen:
  - Listenumschreibung
  - primitiv und nicht-primitiv rekursive Funktionen
- Polymorphie :
  - Abstraktion von Typen durch Typvariablen
- Lokale Definitionen mit `where`
- Überblick: vordefinierte Funktionen auf Listen